query: #------------------------------------------------
  # alle Landkreise und kreisfreie Städte 
  # 294 Landkreise             --> de:Liste_der_Landkreise_in_Deutschland
  # 107 Kreisfreie Städte      --> de:Liste der kreisfreien Städte in Deutschland
  # Summe 401 Stand 2019-03-27 
  #-----------------------------------------------
  #defaultView:Map,Table
  SELECT distinct ?item ?itemLabel ?itemDescription ?kreisschluessel 
         (SAMPLE(?coordinate) AS ?coord)              # nur eine Koordinate soll genutzt werden
  WHERE {
    { ?item wdt:P31/wdt:P279* wd:Q106658.            # alle Landkreise (294 von 294)
      MINUS { ?item wdt:P576 _:b1. }                 # ohne Auflösungsdatum
    }
    union
    {  ?item p:P31 ?statement .
       ?statement ps:P31/wdt:P279* wd:Q22865 .          # alle Kreisfreie Städte in Deutschland  (107 von 107)
       MINUS { ?statement pq:P582 [] }                  # ohne Auflösungsdatum
    }
    
    OPTIONAL { ?item wdt:P440 ?kreisschluessel     }              # Kreise mit Kreisschlüssel (399 von 401)
    MINUS { ?item wdt:P576 _:b2. }                    # ohne Auflösungsdatum   
    OPTIONAL { ?item wdt:P625 ?coordinate. }          # Koordinate für Kartenansicht
    SERVICE wikibase:label { bd:serviceParam wikibase:language "de,en,[AUTO_LANGUAGE]" }
  }
  GROUP BY ?item ?itemLabel ?itemDescription ?coord ?kreisschluessel
  order by ?itemLabel